
Rozdział 16.
Obliczenia różne

W tym rozdziale:

 Przeliczanie jednostek miary

 Rozwiązywanie trójkątów prostokątnych

 Obliczanie pola, powierzchni, obwodu i objętości

 Różne metody zaokrąglania wartości

Ten rozdział zawiera pomocnicze informacje, które w pewnych sytuacjach mogą się okazać bardzo
przydatne. Można potraktować go jako coś w rodzaju ściągawki, która ułatwia przypomnienie
sobie o sprawach kiedyś świetnie znanych, ale które dawno już wyleciały z pamięci.

Przeliczanie jednostek
Chyba każdy zetknął się kiedyś z koniecznością przeliczenia odległości między dwoma miastami
z mil na kilometry (lub na odwrót), na przykład ze względu na wymagania zagranicznego oddziału
firmy. Jaki jest współczynnik takiego przeliczenia?

Funkcja KONWERTUJ służy do przeliczania między różnymi rodzajami jednostek w następujących
kategoriach:

 obszar,

 odległość,

 energia,

 siła,

 informacja,

 pole magnetyczne,

 moc,

 ciśnienie,

 szybkość,

 temperatura,

 czas,

 objętość (miara ilości płynu),

 ciężar i masa.

Kup książkę Poleć książkę

http:/helion.pl/page54976~rt/ex13bi
http:/helion.pl/page54976~rf/ex13bi


322 Część II  Formuły i funkcje

W wersjach Excela starszych od 2007 funkcja KONWERTUJ (pod nazwą CONVERT) wymagała
zainstalowania dodatku Analysis ToolPak. Począwszy od Excela 2007, ta użyteczna funkcja
jest dostępna bezpośrednio w programie.

Funkcja KONWERTUJ przyjmuje trzy argumenty — wartość do przeliczenia, jednostkę źródłową
oraz jednostkę docelową. Jeśli na przykład komórka A1 zawiera odległość wyrażoną w milach,
to za pomocą następującej formuły można przeliczyć ją na kilometry:

=KONWERTUJ(A1;"mi";"km")

Drugi oraz trzeci argument to skróty nazw jednostek opisane w systemie pomocy Excela. Niektóre
z tych skrótów są powszechnie używane i łatwo się ich domyślić. Formuła wymaga oczywiście
podania dokładnego skrótu. Ponadto w skrótach nazw jednostek są rozróżniane wielkie i małe litery,
więc następująca formuła zwróci błąd:

=KONWERTUJ(A1;"Mi";"km")

Funkcja KONWERTUJ jest bardziej uniwersalna, niż mogłoby się wydawać. W przypadku jednostek
metrycznych można użyć prefiksów odpowiadających mnożnikom wartości. Takiego prefiksu
użyłem w pierwszym z podanych przykładów — skrót podstawowej metrycznej jednostki odle-
głości to m (metr), ja zaś dodałem do niego prefiks k (kilo), aby wyrazić wynik w kilometrach.

Czasami przydaje się odrobina kreatywności. Jeśli na przykład należy przeliczyć prędkość 100 km/h
na mile na sekundę, to można użyć w tym celu następującej formuły:

=KONWERTUJ(100;"km";"mi")/KONWERTUJ(1;"hr";"sec")

Funkcja KONWERTUJ w Excelu 2013 została znacząco usprawniona i obsługuje wiele nowych
jednostek.

Rysunek 16.1 przedstawia fragment tabeli, w której są wymienione wszystkie jednostki obsłu-
giwane przez funkcję KONWERTUJ. Tabela ta może być sortowana i filtrowana, ponadto zostały w niej
wyszczególnione jednostki, w których można stosować prefiksy metryczne oraz jednostki, których
obsługa została dodana w Excelu 2013.

Rysunek 16.1. Fragment tabeli z jednostkami obsługiwanymi przez funkcję KONWERTUJ

Potrzeba innych jednostek?

Funkcja KONWERTUJ nie obsługuje wszystkich możliwych jednostek. Aby przeliczyć wartość, która nie
jest przez nią obsługiwana, trzeba znaleźć odpowiedni przelicznik. Internet jest dobrym źródłem tego
typu informacji — na ogół wystarczy wpisać w wyszukiwarce nazwę potrzebnej jednostki, aby znaleźć
odpowiednie dane do przeliczenia.

Polecam też popularny (i darmowy) program Convert do przeliczania jednostek autorstwa Josha Madisona
(http://joshmadison.com/convert-for-windows/). Ten znakomity program obsługuje chyba wszystkie
jednostki miary, jakie kiedykolwiek powstały.

Skoroszyt z jednostkami można pobrać z serwera FTP pod adresem ftp://ftp.helion.pl/
przyklady/ex13bi.zip. Nosi on nazwę tabela konwersji jednostek.xlsx.

Jeżeli funkcja KONWERTUJ nie obsługuje potrzebnej jednostki, to być może jednostka ta jest obsłu-
giwana przez inną, wyspecjalizowaną funkcję Excela. W tabeli 16.1 zgromadzono kilka innych
funkcji służących do przeliczania jednostek miar.

Tabela 16.1. Inne funkcje do przeliczania jednostek

Funkcja Opis

ARABSKIE* Przelicza liczbę rzymską na arabską.

PODSTAWA* Przelicza liczbę dziesiętną na wartość o zadanej podstawie.

DWÓJK.NA.DZIES Przelicza wartość dwójkową na dziesiętną.

DWÓJK.NA.ÓSM Przelicza wartość dwójkową na ósemkową.

DZIES.NA.DWÓJK Przelicza wartość dziesiętną na dwójkową.

DZIES.NA.SZESN Przelicza wartość dziesiętną na szesnastkową.

DZIES.NA.ÓSM Przelicza wartość dziesiętną na ósemkową.

STOPNIE Przelicza kąt w radianach na wartość w stopniach.

SZESN.NA.DWÓJK Przelicza wartość szesnastkową na dwójkową.

SZESN.NA.DZIES Przelicza wartość szesnastkową na dziesiętną.

SZESN.NA.ÓSM Przelicza wartość szesnastkową na ósemkową.

ÓSM.NA.DWÓJK Przelicza wartość ósemkową na dwójkową.

ÓSM.NA.DZIES Przelicza wartość ósemkową na dziesiętną.

ÓSM.NA.SZESN Przelicza wartość ósemkową na szesnastkową.

RADIANY Przelicza kąt w stopniach na wartość w radianach.

*Nowe funkcje Excela 2013

Rozwiązywanie trójkątów prostokątnych
Trójkąt prostokątny można opisać sześcioma własnościami — trzema bokami i trzema kątami.
Rysunek 16.2 przedstawia trójkąt prostokątny z oznaczeniem poszczególnych cech — kąty zo-
stały oznaczone literami A, B oraz C, natomiast boki nazwami przeciwprostokątna, przyprosto-
kątna a oraz przyprostokątna b. Kąt C zawsze wynosi 90° (lub π/2 radianów). Jeśli znana jest
wartość dwóch dowolnych z wymienionych cech (oprócz kąta C, który jest zawsze znany), po-
zostałe można obliczyć z odpowiednich wzorów.

Rysunek 16.2.
Elementy trójkąta
prostokątnego

Zgodnie z twierdzeniem Pitagorasa przyprostokątna a2+przyprostokątna b2=przeciwprostokątna2.
Jeśli znane są długości dwóch dowolnych boków trójkąta, na podstawie tego twierdzenia można
obliczyć trzeci bok. Formuła umożliwiająca obliczenie długości przyprostokątnej b (przy znanej
przeciwprostokątnej i drugiej przyprostokątnej) ma następującą postać:

=PIERWIASTEK((przeciwprostokątna^2)-(przyprostokątnaA^2))

W analogiczny sposób można obliczyć wartość przyprostokątnej a (jeśli znana jest przeciwpro-
stokątna i przyprostokątna b):

=PIERWIASTEK((przeciwprostokątna^2)-(przyprostokątnaB^2))

A oto formuła umożliwiająca obliczenie długości przeciwprostokątnej trójkąta prostokątnego
(przy znanej długości obydwu przyprostokątnych):

=PIERWIASTEK((przyprostokątnaA^2)+(przyprostokątnaB^2))

Inne przydatne funkcje trygonometryczne:

SIN(A) = przyprostokątna b / przeciwprostokątna

SIN(B) = przyprostokątna a / przeciwprostokątna

COS(A) = przyprostokątna a / przeciwprostokątna

COS(B) = przyprostokątna b / przeciwprostokątna

TAN(A) = przyprostokątna b / przyprostokątna a
TAN(B) = przyprostokątna a / przyprostokątna b

We wszystkich funkcjach trygonometrycznych w Excelu przyjęto założenie, że wartości kątów
są podawane w radianach. Aby przeliczyć stopnie na radiany, należy użyć funkcji RADIANY.
Do przeliczania radianów na stopnie służy funkcja STOPNIE.

Jeśli znane są wartości przyprostokątnych, to za pomocą poniższej formuły można obliczyć
wartość kąta pomiędzy przeciwprostokątną a przyprostokątną a (czyli kąta A):

=ATAN(przyprostokątnaB/przyprostokątnaA)

Powyższa formuła zwraca wynik w radianach. Aby przeliczyć go na stopnie, należy użyć nastę-
pującej formuły:

=STOPNIE(ATAN(przyprostokątnaB/przyprostokątnaA))

Jeśli znane są wartości przyprostokątnych, to za pomocą poniższej formuły można obliczyć
wartość kąta pomiędzy przeciwprostokątną a przyprostokątną b (czyli kąta B):

=PI()/2-ATAN(przyprostokątnaB/przyprostokątnaA)

Powyższa formuła zwraca wynik w radianach. Aby przeliczyć go na stopnie, należy użyć nastę-
pującej formuły:

=90-STOPNIE(ATAN(przyprostokątnaB/przyprostokątnaA))

Obliczanie pola, powierzchni, obwodu i objętości
W tej części rozdziału zostały opisane formuły służące do obliczania pola, powierzchni, obwodu
oraz objętości prostych figur geometrycznych oraz brył.

Obliczanie pola oraz obwodu kwadratu
Aby obliczyć pole kwadratu, należy po prostu podnieść długość jego boku do kwadratu. Poniższa
formuła oblicza pole kwadratu na podstawie zawartości komórki o nazwie bok:

=bok^2

Aby obliczyć obwód kwadratu, należy pomnożyć długość jego boku przez 4. Poniższa formuła
wylicza obwód kwadratu na podstawie komórki o nazwie bok:

=bok*4

Obliczanie pola oraz obwodu prostokąta
Aby obliczyć pole prostokąta, należy pomnożyć wartość jego podstawy przez wysokość. Poniższa
formuła zwraca pole prostokąta wyliczone na podstawie zawartości komórek o nazwie wysokość
oraz podstawa:

=wysokość*podstawa

Aby obliczyć obwód prostokąta, należy pomnożyć jego wysokość przez 2 i do rezultatu dodać
szerokość pomnożoną przez 2. Poniższa formuła zwraca pole prostokąta wyliczone na podstawie
zawartości komórek o nazwie wysokość oraz podstawa:

=(wysokość*2)+(szerokość*2)

Obliczanie pola oraz obwodu koła
Aby obliczyć pole koła, należy pomnożyć kwadrat jego promienia przez wartość liczby π. Poniż-
sza formuła zwraca pole koła wyliczone na podstawie promienia zawartego w komórce o nazwie
promień:

=PI()*(promień^2)

Promień koła jest równy połowie jego średnicy.

Aby obliczyć obwód koła, należy pomnożyć jego średnicę przez wartość liczby π. Poniższa
formuła zwraca obwód koła wyliczony na podstawie wartości średnicy podanej w komórce
o nazwie średnica:

=średnica*PI()

Średnica koła jest równa dwukrotności promienia.

Obliczanie pola trapezu
Aby obliczyć pole trapezu, należy zsumować długości jego podstaw, pomnożyć tę sumę przez
wysokość i podzielić otrzymany wynik przez 2. Poniższa formuła zwraca wartość pola trapezu
wyliczoną na podstawie wartości w komórkach podstawa1, podstawa2 oraz wysokość:
=((podstawa1+podstawa2)*wysokość)/2

Obliczanie pola trójkąta
Aby obliczyć pole trójkąta, należy pomnożyć długość jego podstawy przez wysokość i podzielić
otrzymany wynik przez 2. Poniższa formuła zwraca wartość pola trójkąta wyliczoną na podsta-
wie wartości w komórkach podstawa oraz wysokość:
=(podstawa*wysokość)/2

Obliczanie powierzchni oraz objętości kuli
Aby obliczyć powierzchnię kuli, należy pomnożyć kwadrat jej promienia przez π, a następnie
pomnożyć wynik przez 4. Poniższa formuła zwraca powierzchnię kuli, wyliczoną na podstawie
wartości promienia w komórce o nazwie promień:

=PI()*(promień^2)*4

Aby obliczyć objętość kuli, należy pomnożyć sześcian promienia przez 4*π, natomiast otrzymany
rezultat podzielić przez 3. Poniższa formuła oblicza objętość kuli na podstawie wartości promienia
w komórce o nazwie promień:

=((radius^3)*(4*PI()))/3

Obliczanie powierzchni oraz objętości sześcianu
Aby obliczyć powierzchnię sześcianu, należy podnieść do kwadratu długość jednego boku i po-
mnożyć wynik przez 6. Poniższa formuła wylicza powierzchnię sześcianu na podstawie długości
boku w komórce o nazwie bok:

=(bok^2)*6

Aby obliczyć objętość sześcianu, należy podnieść długość jego boku do potęgi trzeciej. Poniższa
formuła zwraca objętość sześcianu wyliczoną na podstawie długości boku w komórce o nazwie bok:

=bok^3

Obliczanie powierzchni oraz objętości prostopadłościanu
Poniższa formuła oblicza powierzchnię prostopadłościanu na podstawie wartości zapisanych
w komórkach o nazwach wysokość, szerokość i długość:
=(długość*wysokość*2)+(długość*szerokość*2)+(szerokość*wysokość*2)

Aby obliczyć objętość prostopadłościanu, należy pomnożyć jego wysokość przez szerokość i długość:

=wysokość*szerokość*długość

Obliczanie powierzchni oraz objętości stożka
Poniższa formuła oblicza powierzchnię stożka (z uwzględnieniem powierzchni podstawy). Do wyli-
czeń używane są wartości wprowadzone w komórkach o nazwach promień oraz wysokość:
=PI()*promień*(PIERWIASTEK(wysokość^2+promień^2)+promień))

Aby obliczyć objętość stożka, należy pomnożyć kwadrat promienia przez π, otrzymany wynik
pomnożyć przez wysokość i podzielić przez 3. Poniższa formuła oblicza objętość stożka na pod-
stawie wartości w komórkach o nazwach promień oraz wysokość:
=(PI()*(promień^2)*wysokość)/3

Obliczanie objętości walca
Aby obliczyć objętość walca, należy pomnożyć kwadrat promienia podstawy przez π, a następ-
nie pomnożyć wynik przez wysokość. Poniższa formuła oblicza objętość walca na podstawie
wartości w komórkach o nazwach promień oraz wysokość:
=(PI()*(promień^2)*wysokość)

Obliczanie objętości graniastosłupa
Należy pomnożyć pole podstawy przez wysokość, a następnie podzielić wynik przez 3. Poniższa
formuła oblicza objętość graniastosłupa przy założeniu, że w komórkach szerokość (szerokość
podstawy), długość (długość podstawy) oraz wysokość (wysokość bryły) wprowadzono odpo-
wiednie wartości:

=(szerokość*długość*wysokość)/3

Zaokrąglanie liczb
Excel oferuje wiele funkcji służących do zaokrąglania liczb na różne sposoby. Ich zestawienie
zawiera tabela 16.2.

Należy pamiętać, że zaokrąglenie wartości to nie to samo co jej sformatowanie. Po sformatowaniu
wartości polegającym na wyświetleniu określonej liczby miejsc dziesiętnych formuły odwołujące się
do tej wartości nadal wykorzystują jej pełną wartość, która może się różnić od wyświetlonej. Z kolei
po zaokrągleniu liczby odwołujące się do niej formuły będą korzystały z zaokrąglonej wartości.

Tabela 16.2. Funkcje służące do zaokrąglania liczb w Excelu

Funkcja Opis

ZAOKR.W.GÓRĘ Zaokrągla wartość w górę (od zera) do najbliższej podanej wielokrotności.

CENA.DZIES Przelicza cenę wyrażoną w postaci ułamkowej na liczbę dziesiętną.

CENA.UŁAM Przelicza cenę wyrażoną w postaci dziesiętnej na liczbę ułamkową.

ZAOKR.DO.PARZ Zaokrągla wartość w górę (od zera) do najbliższej parzystej liczby całkowitej.

ZAOKR.W.DÓŁ Zaokrągla wartość w dół (do zera) do najbliższej podanej wielokrotności.

ZAOKR.DO.CAŁK Zaokrągla wartość w dół do najbliższej liczby całkowitej.

ISO.ZAOKR.W.GÓRĘ* Zaokrągla numer w górę do najbliższej liczby całkowitej albo do najbliższej wielokrotności
podanej wartości (istotności). Podobna do ZAOKR.W.GÓRĘ, ale przyjmuje także argumenty
ujemne.

ZAOKR.DO.WIELOKR Zaokrągla wartość do podanej wielokrotności.

ZAOKR.DO.NPARZ Zaokrągla wartość w górę (od zera) do najbliższej nieparzystej liczby całkowitej.

ZAOKR Zaokrągla wartość do podanej liczby cyfr.

ZAOKR.DÓŁ Zaokrągla wartość w dół (do zera) do podanej liczby cyfr.

ZAOKR.GÓRA Zaokrągla wartość w górę (od zera) do podanej liczby cyfr.

LICZBA.CAŁK Obcina liczbę do liczby całkowitej lub do podanej dokładności.

*Funkcja wprowadzona w Excelu 2010

W tej części rozdziału podano przykłady formuł wykorzystujących różne metody zaokrąglania.

Proste formuły do zaokrąglania
Funkcja ZAOKR przydaje się do wykonywania prostych operacji zaokrąglania z uwzględnieniem
określonej liczby miejsc po przecinku. Liczbę miejsc podaje się w drugim argumencie funkcji
ZAOKR. Na przykład poniższa formuła zwraca wartość 123,40 (wartość została zaokrąglona do
jednego miejsca po przecinku):

=ZAOKR(123,37;1)

Jeśli drugim argumentem funkcji ZAOKR będzie zero, wartość jest zaokrąglana do najbliższej liczby
całkowitej. Poniższa formuła zwraca wartość 123,00:

=ZAOKR(123,37;0)

Drugi argument funkcji ZAOKR może mieć wartość ujemną. W takim przypadku wartość jest
zaokrąglana do kolejnych miejsc przed przecinkiem. Na przykład poniższa formuła zwraca
wartość 120,00:

=ZAOKR(123,37;-1)

Funkcja ZAOKR może zaokrąglać w górę albo w dół. Ale w jaki sposób poradzi sobie z wartością
taką jak 12,5, zaokrągloną do pełnej liczby całkowitej? Okazuje się, że funkcja ZAOKR zaokrągla
takie wartości „od zera”. Na przykład poniższa formuła zwraca wartość 13,0:

=ZAOKR(12,5;0)

Kolejna formuła zwraca wartość -13,00 (gdyż zaokrąglenie zawsze przebiega „od zera”):

=ZAOKR(-12,5;0)

Aby wymusić zaokrąglanie w określonym kierunku, należy użyć funkcji ZAOKR.GÓRA albo
ZAOKR.DÓŁ. Na przykład poniższa formuła zwraca wartość 12,0 (wartość źródłowa została za-
okrąglona w dół).

=ZAOKR.DÓŁ(12,5;0)

Następująca formuła zwraca wartość 13,0. Podany argument został zaokrąglony do najbliższej
pełnej liczby całkowitej.

=ZAOKR.DÓŁ(12,43;0)

Zaokrąglanie do najbliższej wielokrotności
Funkcja ZAOKR.DO.WIELOKR przydaje się do zaokrąglania wartości do najbliższej wielokrotności.
Można jej użyć na przykład do zaokrąglenia wartości do wielokrotności liczby 5. Poniższa formuła
zwraca wartość 135:

=ZAOKR.DO.WIELOKR(133;5)

Zaokrąglanie wartości walutowych
Często przydaje się zaokrąglanie wartości walutowych — na przykład w sytuacji, gdy wartość w zło-
tych trzeba zaokrąglić do pełnego grosza, jeśli wyliczona cena produktu wynosi np. 45,78923 zł. Brzmi
to dość prosto, ale wbrew pozorom taką operację można wykonać na trzy sposoby:

 Zaokrąglić cenę w górę do pełnego grosza.

 Zaokrąglić cenę w dół do pełnego grosza.

 Zaokrąglić cenę do pełnego grosza (w górę albo w dół).

W poniższej formule przyjęto założenie, że w komórce A1 znajduje się wartość wyrażona w zło-
tych i w groszach. Gdyby wynosiła ona na przykład 12,421 zł, to poniższa formuła zwróciłaby
wartość 12,42 zł.

=ZAOKR(A1;2)

Kup książkę Poleć książkę

http:/helion.pl/page54976~rt/ex13bi
http:/helion.pl/page54976~rf/ex13bi


Rozdział 16.  Obliczenia różne 329

Aby zaokrąglić cenę w górę do pełnego grosza, należy użyć funkcji ZAOKR.W.GÓRĘ. Poniższa for-
muła zaokrągla wartość w komórce A1 w górę, do pełnego grosza. Jeżeli na przykład w komórce A1
znajdowałaby się wartość 12,421 zł, to poniższa formuła zwróciłaby rezultat 12,43 zł.

=ZAOKR.W.GÓRĘ(A1;0,01)

Aby zaokrąglić kwotę w dół, należy użyć funkcji ZAOKR.W.DÓŁ. Na przykład poniższa formuła
zaokrągla wartość w komórce A1 w dół, do pełnego grosza. Gdyby w komórce A1 znajdowała
się wartość 12,421 zł, to poniższa formuła zwróciłaby rezultat 12,42 zł.

=ZAOKR.W.DÓŁ(A1;0,01)

Aby zaokrąglić kwotę w górę do pełnych pięciu groszy, należy użyć poniższej formuły:

=ZAOKR.W.GÓRĘ(A1;0,05)

W handlu bardzo wiele cen kończy się na 0,99 zł. Aby całkowitą kwotę w złotych zastąpić taką,
która ma końcówkę 99-groszową, należy po prostu odjąć 0,01 zł od końcowej ceny. Niektóre
droższe produkty kończą się wartością 9,99 zł. Aby „zaokrąglić” kwotę do 9,99 zł, należy naj-
pierw zaokrąglić ją do pełnych 10 zł, a potem odjąć jeden grosz. Jeżeli w komórce A1 znajduje się
cena, to poniższa formuła przekształci ją na taką, która kończy się na 9,99 zł.

=ZAOKR(A1/10;0)*10-0,01

Jeśli na przykład w komórce A1 znajdowałaby się kwota 345,78 zł, to formuła zwróciłaby war-
tość 349,99 zł.

Podobny efekt można uzyskać prościej za pomocą funkcji ZAOKR.DO.WIELOKR:

=ZAOKR.DO.WIELOKR(A1;10)-0,01

Obsługa kwot ułamkowych
Funkcje CENA.UŁAM oraz CENA.DZIES przydają się przy przekształcaniu ułamkowych wartości kwot,
spotykanych m.in. w kwotowaniach papierów wartościowych.

Weźmy na przykład wartość 9,25 zł. Część ułamkową tej ceny można wyrazić w postaci ułam-
ka zwykłego (9 1/4 zł, 9 2/8 zł, 9 4/16 itp.). Funkcja CENA.UŁAM przyjmuje dwa argumenty — war-
tość w złotych oraz mianownik wartości ułamkowej. Na przykład poniższa formuła zwraca
wartość 9,1 (część dziesiętna ,1 odpowiada ułamkowi 1/4):

=CENA.UŁAM(9,25;4)

W większości przypadków nie należy używać wartości zwracanych przez funkcję CENA.UŁAM do
innych obliczeń. Na przykład wartość wyliczona w poprzednim przykładzie zostałaby potraktowana
w kolejnych obliczeniach jako 9,1, a nie 9,25. Aby przeprowadzić obliczenia na takiej wartości,
trzeba ją z powrotem przekształcić na postać dziesiętną za pomocą funkcji CENA.DZIES.

Funkcja CENA.DZIES przekształca wartość w złotych wyrażoną w postaci ułamkowej na zwykłą
wartość w notacji dziesiętnej. Przyjmuje ona ponadto drugi argument w postaci mianownika
części ułamkowej. Poniższa formuła zwraca wartość 9,25:

=CENA.DZIES(9,1;4)

Zastosowanie funkcji CENA.DZIES oraz CENA.UŁAM nie ogranicza się do kwot. Można jej użyć
na przykład do przeliczania stóp i cali. Przypuśćmy, że dana jest wartość 8½ stopy. Przy użyciu poniższej
formuły można ją wyrazić w stopach i calach. Zwraca ona wartość 8,06 (co oznacza 8 stóp, 6 cali).

=CENA.UŁAM(8,5;12)

Znawcy baseballu mogą wykorzystać ją do obliczania statystyk. Rozegranie przez miotacza
(ang. pitcher), średnio 62/3 zmiany jest zapisywane w uproszczony sposób jako 6,2.
Poniższa formuła zwraca dokładnie taki wynik:

=CENA.UŁAM(6+2/3;3)

Zastosowanie funkcji ZAOKR.DO.CAŁK oraz LICZBA.CAŁK
Funkcje ZAOKR.DO.CAŁK oraz LICZBA.CAŁK wydają się podobne. Obydwie zamieniają podaną war-
tość na liczbę całkowitą, z tym że funkcja LICZBA.CAŁK dokonuje tego poprzez obcięcie części
ułamkowej, natomiast funkcja ZAOKR.DO.CAŁK poprzez zaokrąglenie w dół, do najbliższej warto-
ści całkowitej, na bazie wartości części ułamkowej.

W praktyce funkcje ZAOKR.DO.CAŁK oraz LICZBA.CAŁK zwracają różne wartości tylko w przypadku
ujemnych argumentów. Na przykład następująca formuła zwraca wartość -14,0.

=LICZBA.CAŁK(-14;2)

Następna formuła zwraca wartość -15, gdyż –14,3 zostaje zaokrąglone w dół, do kolejnej,
mniejszej liczby całkowitej.

=ZAOKR.DO.CAŁK(-14;2)

Funkcja LICZBA.CAŁK przyjmuje jeden dodatkowy (opcjonalny) argument, który można wyko-
rzystać do obcinania nadmiarowych miejsc po przecinku. Na przykład poniższa formuła zwraca
wartość 54,33 (argument został obcięty do dwóch miejsc po przecinku).

=LICZBA.CAŁK(54,3333333;2)

Zaokrąglanie do parzystych i nieparzystych liczb całkowitych
Funkcje ZAOKR.DO.NPARZ oraz ZAOKR.DO.PARZ przydają się w sytuacjach, w których należy zaokrąglić
liczbę w górę, do najbliższej nieparzystej albo parzystej liczby całkowitej. Funkcje te przyjmują
jeden argument i zwracają wartość całkowitą. Funkcja ZAOKR.DO.PARZ zaokrągla argument w górę
do najbliższej liczby parzystej. Funkcja ZAOKR.DO.NPARZ zaokrągla argument w górę, do najbliższej
liczby nieparzystej. W tabeli 16.3 pokazano kilka przykładów działania tych funkcji.

Tabela 16.3. Rezultaty działania funkcji ZAOKR.DO.NPARZ oraz ZAOKR.DO.PARZ

Wartość Funkcja ZAOKR.DO.PARZ Funkcja ZAOKR.DO.NPARZ

–3,6 –4 –5

–3,0 –4 –3

–2,4 –4 –3

–1,8 –2 –3

–1,2 –2 –3

–0,6 –2 –1

0,0 0 1

0,6 2 1

1,2 2 3

1,8 2 3

2,4 4 3

3,0 4 3

3,6 4 5

Zaokrąglanie do n cyfr znaczących
W niektórych przypadkach przydaje się możliwość zaokrąglenia wartości do konkretnej liczby
cyfr znaczących. Na przykład wartość 1 432 187 zaokrąglona do dwóch cyfr znaczących będzie
miała wartość 1 400 000, natomiast wartość 9 187 877 po zaokrągleniu do trzech cyfr znaczą-
cych będzie wynosiła 9 180 000.

Jeśli wartość źródłowa jest liczbą dodatnią, bez części dziesiętnej, to można w tym celu użyć
formuły podanej niżej. Ta formuła zaokrągla wartość w komórce A1 do dwóch cyfr znaczących.
Aby zaokrąglić argument do innej liczby cyfr znaczących, należy zastąpić wartość 2 w poniższej
formule inną cyfrą.

=ZAOKR.DÓŁ(A1;2-DŁ(A1))

W przypadku wartości niebędących liczbami całkowitymi rozwiązanie jest trochę bardziej skom-
plikowane. Poniższa formuła to ogólny przepis na zaokrąglanie wartości zawartej w komórce A1
do liczby cyfr znaczących podanej w komórce A2. Tę formułę można zastosować w przypadku
dodatnich i ujemnych liczb całkowitych oraz niecałkowitych.

=ZAOKR(A1;A2-1-ZAOKR.DO.CAŁK(LOG10(MODUŁ.LICZBY(A1))))

Jeśli na przykład w komórce A1 wprowadzono wartość 1,27845, a wartość w komórce A2 wynosi 3,
to formuła zwróci wartość 1,28000 (czyli argument zaokrąglony do trzech cyfr znaczących).

